
1 Introduction

The Hirota direct method was first published in a paper by Hirota in 1971. The in-
troduction of this approach provided a direct method for finding N-soliton solutions to
non-linear evolutionary equations and, by way of an example, Hirota applied this method
to the Korteweg-deVries (KdV) equation [Hir-71], i.e.

ut + 6uux + uxxx = 0, (1)

where u = u(x, t).

As part of Hirota’s analysis he introduced a bilinear form defined as

Dx(f · g) =

(
∂

∂x
− ∂

∂x′

)
f(x) · g(x′)

∣∣∣∣
x′=x

, (2)

or, more generally 2

Dm
x D

n
t (f · g) =

(
∂

∂x
− ∂

∂x′

)m(
∂

∂t
− ∂

∂t′

)n
f(x) · g(x′)

∣∣∣∣
x′=x, t′=t

, (3)

where D is a binary operator (because it operates on a pair of functions) and is called
the Hirota derivative.

On expanding an mth power of the bilinear operator of eqn. (2), we see that it results
in a binomial expression of the form(

∂

∂x
− ∂

∂x′

)m
=

m∑
r=0

(−1)r
(
m

r

)
∂rx∂

m−r
x′ =

m∑
i=0

(−1)r
(
m

r

)
∂m−rx ∂mx′ , (4)

where we have made use of the binomial coefficient,
(
m
r

)
=

m!

r!(m− r)
, 0 ≤ r ≤ m. This,

in turn, gives rise to a more compact form for the Hirota derivative, i.e.
1email: graham@griffiths1.com
2Note: The Hirota derivative acts on a product of two functions in a similar way to the standard Calculus rule for

differentiation. However, there is a sign difference, e.g. Dx(f · g) ≡ fxg − fgx compared with ∂x(f g) ≡ fxg + fgx.

Dm
x f · g =

m∑
r=0

(−1)i
(
m

r

)
f(m−r)x · grx . (5)

This form is particularly useful for a computer algebra system (CAS) implementation, as
we see below.

Some examples of bilinear operations follow.

Dx(f · g) = fxg − fgx = −Dx(g · h)
D2
x(f · g) = fxxg − 2fxgx + fgxx = D2

x(g · f)
DxDt(f · g) = Dx(ftg − fgt) = fxtg − ftgx − fxgt + fgxt = DxDt(g · f)
DxDt(f · f) = 2(ffxt − fxft)
D4
x(f · g) = fxxxxg − 4fxxxgx + 6fxxgxx − 4fxgxxx + fgxxxx = D4

x(g · f)
Dn
x(f · g) = 0, (n odd)

(6)

Also, fundamental to Hirota’s approach was the introduction of the ancillary function
f = f(x, t), which is a logarithmic transformation defined as

u(x, t) = 2
∂2 ln f(x, t)

∂x2
= 2

ffxx − f 2
x

f 2
, (7)

where the leading constant of 2 was chosen so that, on application of this equation,
certain terms cancel to yield a quadratic in f .

The form of function f in eqn. (7) is chosen to be a perturbation expansion, defined as

f(x, t) = 1 +
∞∑
n=1

εnfn(x, t), (8)

for functions f1, f2, · · · , etc. yet to be found.

Now, for a given bilinear operation, say B, where we have B(f · f) = 0, the coefficient of
like powers of epsilon can be equated to zero to obtain the following set of equations,

O(ε0) :B(1 · 1) = 0, (9)

O(ε1) :B(1 · f1 + f1 · 1) = 0, (10)

O(ε2) :B(1 · f2 + f1 · f1 + f2 · 1) = 0, (11)

O(ε3) :B(1 · f3 + f1 · f2 + f2 · f1 + f3 · 1) = 0, (12)

O(ε4) :B(1 · f4 + f1 · f3 + f2 · f2 + f3 · f1 + f4 · 1) = 0, (13)

O(εn) :B

(
n∑
j=0

fj · fn−j

)
= 0, (f0 = 1). (14)

The above scheme is general and is applicable to any explicit bilinear operator expression.
It can be shown that if the original PDE admits a N -soliton solution, then eqn. (8) will
truncate at the n = N term provided f1 is the sum of precisely N simple exponential
terms. This leads to

f(x, t) = 1 + ε1f1 :one soliton, (15)

f(x, t) = 1 + ε1f1 + ε2f2 :two solitons, (16)

f(x, t) = 1 + ε1f1 + ε2f2 + ε3f3 :three solitons, (17)

etc. (18)

We will now illustrate use of eqn. (10) by application of the bilinear examples in eqns.
(6) above, when we see that for m = 1 and n = 1,

DxDt(1 · f1) = Dx(−1 · f1,t) = f1,xt, (19)

DxDt(f1 · 1) = Dx(f1,t · 1) = f1,xt, (20)

∴ DxDt(1 · f1 + f1 · 1) = 2f1,xt. (21)

Similarly, for m = 4 and n = 0 we see that

D4
x(1 · f1) = f1,xxxx, (22)

D4
x(f1 · 1) = f1,xxxx, (23)

∴ D4
x(1 · f1 + f1 · 1) = 2f1,xxxx. (24)

Thus, we have (
DxDt +D4

x

)
(1 · f1 + f1 · 1) = 2(f1,xt + f1,xxxx). (25)

A similar approach to eqn (11) yields,

(
DxDt +D4

x

)
(1 · f2 + f1 · f1 + f2 · 1) = 2(f1f1,xt − f1,xf1,t)+

2
[
(f2,xt + f2,xxxx) + (f1f1,xxxx − 4f1,xf1,xxx + 3f 2

1,xx)
]
. (26)

We make use of these relationships to obtain appropriate dispersion relations and coupling
coefficients in the KdV examples below.

Whilst the above equations look quite complex, they do provide a route to solving evo-
lutionary equations. This will be illustrated by obtaining detailed solutions to the above
KdV equation (1) by way of example. In addition, solutions to other well known evolu-
tionary equations will be presented.

Additional expanded discussion on this method can be found in [Dra-92], [Hir-04], [Her-94]
and [Waz-09].

2 Applications of the Hirota method

2.1 Single-soliton solution to the KdV equation

The KdV equation (1) which we repeat here for convenience,

ut + 6uux + uxxx = 0, (27)

can be transformed by eqn (7), the so-called Cole-Hopf transformation, to yield

2

f 3

[
ft,3xf

2 − 2f3xftf − 4fxft,xf + 4fx
2ft + 4f3xfx,xf−

12f3x
2fx + 16f3xfx

2 + 2 f5xf
2 − 10f4xfxf

]
= 0. (28)

This equation can be written more succinctly in the following form using Hirota’s bilinear
operator,

∂

∂x

[
1

f 2

(
DxDt +D4

x

)
f · f

]
= 0. (29)

Hence, integrating once with respect to x and taking the integration functions to be zero,
the bilinear equation equivalent to eqn. (27) is(

DxDt +D4
x

)
f · f = 0. (30)

Thus, if f is a solution of eqn. (30), then u from eqn. (7) is a solution to eqn. (27). See
[Rog-82, p69-71] for a detailed discussion and derivation.

The problem has now changed to finding the auxiliary function f that satisfies eqn. (30)
which, when inserted into eqn. (7), will yield u(x, t). If we try the ansatz f = 1 and
substitute this into eqn. (7), it leads to the trivial solution u = 0. However, if we try the
ansatz f1 = eξ, where ξ = k(x − ct), k = wavenumber and c = wave velocity, then on
substituting f = 1 + εf1 into eqn. (7), we obtain the solution

u =
2k2e(k(x−ct)

(1 + e(k(x−ct))
2 =

1

2
k2sech2

[
1

2
k(x− ct)

]
, (31)

where we have set ε = 1, without loss of generality.

We can simplify this solution further by substituting f1 into eqn. (10), i.e.

(DxDt +Dxxxx) (1 · f1 + f1 · 1) = 0, (32)

∴ 2
(
−k2c+ k4

)
ek(x−ct) = 0, (33)

where we have used B = (DxDt +Dxxxx) along with eqns. (21) and (24). This leads
directly to the dispersion relation3 for the KdV equation,

c = k2. (34)

Thus, we observe that c = c(k) and k cannot be determined independently for the KdV
equation. With this result, we have finally arrived at the situation where the complete
solution is known, i.e.

u =
1

2
k2sech2

[
1

2
k(x− k2t)

]
. (35)

This is the well known single-soliton solution for the KdV equation. A plot of this solution
is given in Fig (1).

Manual implementation of the above calculations can be rather tedious and is prone
to error, so employing a symbolic computer program is highly recommended. A Maple
program that solves this single-soliton example and generates an animation is given in
Listing (1).

Listing 1: Maple program that uses the Hirota method to derive a single-soliton solution to KdV eqn (27)

# Hirota bilinear method

# Single -soliton solution to KdV equation

restart; with(PDEtools): with(PolynomialTools):

with(plots):

alias(u=u(x,t), v=v(x,t), f=f(x,t)):

# KdV equation

pde1 := diff(u,t)+6*u*diff(u,x)+diff(u,x,x,x);

3The dispersion relation characterizes the mathematical relationship between temporal and spacial variations. For

example, for a harmonic wave of the form A sin

[
2π

λ
(x− ct)

]
we have wavenumber k =

2π

λ
and frequency ω =

2πc

λ
, where

λ represents wavelength, giving the dispersal relation ω(k) = ck.

# Define Hirota Derivatives

Hirota_Dxm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,x$(m-r))

*diff(g,x$r),r=0..m);

Hirota_Dtm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,t$(m-r))

*diff(g,t$r),r=0..m);

Hirota_Dxt :=(f,g)->f*diff(g,x,t)-diff(f,x)*diff(g,t)

-diff(f,t)*diff(g,x)+diff(f,x,t)*g;

v:=2* diff(ln(f),x,x);

declare(f):

pde2:=eval(subs(u=v,pde1))=0;

# Check bilinear transform correct

pde3:= simplify(pde2 ,size);

pde3_chk := simplify(diff(( Hirota_Dxt(f,f)+Hirota_Dxm(f,f,4))/(f^2),x),size)=0;

testeq(lhs(pde3)=lhs(pde3_chk));

# Ansatz for ’f_1 ’

f1:=exp(k*(x-c*t));

B1:=( ff1)->Hirota_Dxt (1,ff1)+Hirota_Dxt(ff1 ,1)+

Hirota_Dxm (1,ff1 ,4)+Hirota_Dxm(ff1 ,1,4)=0;

eqn1:=B1(f1); # B(1.f1 + f1.1);

sol1:=solve(simplify(eqn1 ,symbolic),c);

c:=sol1;

# Single soliton ancilliary solution

F:=1+f1;

# Single soliton solution

soln:=u=normal (2* diff(ln(F),x,x));

# Check solution solution

pdetest(soln ,pde1);

# Animate result

zz1:=subs({k=sqrt (2)},rhs(soln));

animate(zz1 ,x=-50..50,t=-20..20,

numpoints =300, frames =50, axes=framed ,

labels =["x","u(x,t)"],thickness =3,

title="KdV Equation: Single -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16]);

Figure 1: Initial animation profile for KdV equation solution, u(x, t) =
1

2
k2sech2

[
1
2
k(x− k2t)

]
, at t = −20,

for k = 1. The soliton moves left to right.

2.2 Two-soliton solution to the KdV equation

This problem is similar to the single-soliton problem except that the auxilliary function
f has an additional term, so we now need to find f1 and f2. Because we are seeking
a two-soliton solution, we use a two-term form of f1 that was successful for the single
soliton case, i.e. f1 = eξ1 + eξ2 , along with the ansatz f2 = a12e

(ξ1+ξ2), where a12 is

a coupling constant yet to be determined. This enables us to construct the auxilliary
function, f = 1 + εf1 + ε2f2.

The form for f2 is not an arbitrary choice but, rather, is suggested by eqn (11) which
evaluates to

B(1 · f2 + f2.1) +B(f1 · f1) = B(1 · f2 + f2 · 1)− 6k1k2 (k1 − k2)2 eξ1+ξ2 = 0. (36)

For eqn (36) to be equal to zero, clearly, the first term must be of the form,

B(1 · f2 + f2 · 1) = a12e
ξ1+ξ2 . (37)

We now substitute the auxilliary function f into eqn. (7) to obtain the solution

u =
2
(
k2

1e
ξ1 + k2

2e
ξ2 + a12(k1 + k2)2e(ξ1+ξ2)

)
(1 + eξ1 + eξ2 + a12e(ξ1+ξ2))

,

−
2
(
k1e

ξ1 + k2e
ξ2 + a12(k1 + k2)e(ξ1+ξ2)

)2

(1 + eξ1 + eξ2 + a12e(ξ1+ξ2))
2 . (38)

Where, again without loss of generality, we have set ε = 1.

As for the single-soliton solution, we can obtain the dispersion relations by substituting,
in turn, the 1st and 2nd terms of f1 into eqn. (10), to obtain

c1 = k2
1, c2 = k2

2. (39)

Also, on substituting f1 and f2 into eqn. (11) we obtain an expression which can now be
solved for the coupling constant in terms of wave numbers, k1 and k2. This yields,

a12 =
(k1 − k2)2

(k1 + k2)2
. (40)

We now have a fully defined two-soliton solution for KdV eqn (27). A plot of this solution
is given in Fig (2).

A Maple program that solves this two-soliton example and generates an animation is
given in Listing (2).

Listing 2: Maple program that uses the Hirota method to derive a two-soliton solution to KdV eqn (27)

# Hirota bilinear method

# Two -soliton Solution to KdV equation

restart; with(PDEtools): with(PolynomialTools):

with(plots):

alias(u=u(x,t), v=v(x,t), f=f(x,t)):

# KdV equation

pde1 := diff(u,t)+6*u*diff(u,x)+diff(u,x,x,x);

# Convert to potential form

pde2 := subs(u=diff(v,x), pde1);

# Integrate once wrt x

pde3:=int(pde2 ,x);

# Define Hirota Derivatives

Hirota_Dxm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,x$(m-r))

*diff(g,x$r),r=0..m);

Hirota_Dtm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,t$(m-r))

*diff(g,t$r),r=0..m);

Hirota_Dxt :=(f,g)->f*diff(g,x,t)-diff(f,x)*diff(g,t)

-diff(f,t)*diff(g,x)+diff(f,x,t)*g;

v1:=2* diff(ln(f),x); # equivalent to ==>> u=2* diff(ln(f),x,x)

declare(f):

pde4:=eval(subs(v=v1,pde3))=0;

# Check bilinear transform correct

pde5:= simplify(pde4*(f^2/2) ,size);

pde5_chk :=( Hirota_Dxt(f,f)+Hirota_Dxm(f,f,4))/2=0;

testeq(lhs(pde5)=lhs(pde5_chk));

# Ansatz for ’f_1 ’

f11:=exp(k[1]*(x-c[1]*t)); f12:=exp(k[2]*(x-c[2]*t));

f1:=f11+f12;

B1:=( ff1)->Hirota_Dxt (1,ff1)+Hirota_Dxt(ff1 ,1)+

Hirota_Dxm(ff1 ,1,4)+Hirota_Dxm (1,ff1 ,4)=0;

eqn1a :=B1(f11); # B(1.f1 + f1.1);

eqn1b :=B1(f12); # B(1.f2 + f2.1);

sol1:=solve(simplify ({eqn1a ,eqn1b},symbolic) ,{c[1],c[2]});

assign(sol1); # ==>> Dispersion relations

# Ansatz for ’f2 ’ suggested by equation: B(1.f2+f1.f1+f2.1)=0

f2:=a[12]* combine(f11*f12);

# Two -soliton ancilliary function

F:=1+f1+f2;

B2:=(ff1 ,ff2)->Hirota_Dxt (1,ff2)+Hirota_Dxt(ff2 ,1)+

Hirota_Dxm (1,ff2 ,4)+Hirota_Dxm(ff2 ,1,4)+

Hirota_Dxt(ff1 ,ff1)+Hirota_Dxm(ff1 ,ff1 ,4);

eqn2:= simplify(B2(f1,f2),size)=0;

a[12]:= factor(simplify(solve(eqn2 ,a[12]),symbolic));

# Two -soliton solution

soln:=u=simplify (2* diff(ln(F),x,x),size);;

# Check solution

pdetest(soln ,pde1);

# Animate result

zz1:=subs({k[1]=1,k[2]= sqrt (2)},rhs(soln));

animate(zz1 ,x=-50..50,t=-20..20,

numpoints =300, frames =100, axes=framed ,

labels =["x","u(x,t)"],thickness =3,

title="KdV Equation: Two -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16]);

Figure 2: Initial animation profile for KdV equation two-soliton solution, eqn. (38), at t = −20, for k = 1
and k2 =

√
2. Solitons move left to right with the taller faster soliton overtaking the shorter slower soliton.

2.3 Three-soliton solution to the KdV equation

This problem is similar to the two-soliton problem except that the auxilliary function f
has an additional term, so we now need to find f1, f2 and f3. Because we are seeking a
three-soliton solution, we use three-term forms of f1 and f2 that were successful for the
two-soliton case, i.e. f1 = eξ1 + eξ2 + eξ3 and f2 = a12e

(ξ1+ξ2) + a13e
(ξ1+ξ3) + a23e

(ξ2+ξ3),
along with the ansatz f3 = b123e

ξ1+ξ2+ξ3 , where coupling constants a12, a13, a23 and b123

are yet to be determined. This enables us to construct the auxilliary function, f =
1 + εf1 + ε2f2 + ε3f3.

The ai,j auxillary coupling constants are determined as for the two-soliton case. On
substituting in turn all combinations of two terms from f2, i.e. 1st and 2nd; 1st and 3rd;
2nd and 3rd; into eqn. (11), we obtain expressions which can be solved for these coupling
constants in terms of wave numbers, k1, k2 and k3.

Similarly to f2 in the two-soliton case, the form for f3 is not an arbitrary choice but,
rather, is suggested by eqn (12) which evaluates to

B(1 · f3 + f3 · 1)+B(f1 · f1 + f2 · f2) = B(1 · f3 + f3 · 1)−
6 (k2 − k3)2 (k1 − k3)2 (k1 − k2)2 (k1 + k2 + k3)

(k1 + k2) (k1 + k3) (k2 + k3)
eξ1+ξ2+ξ3 = 0. (41)

For eqn (41) to be equal to zero, clearly, the first term must be of the form,

B(1 · f3 + f3 · 1) = b123e
ξ1+ξ2+ξ3 . (42)

We now substitute the auxilliary function f into eqn. (7) to obtain the solution

u =
N1

D1

− 2

(
N2

D2

)2

, (43)

where

N1 =2a12 (k1 + k2)2 eξ1+ξ2 + 2a13 (k1 + k3)2 eξ1+ξ3 + 2a23 (k2 + k3)2 eξ2+ξ3+

2k1
2eξ1 + 2k2

2eξ2 + 2k3
2eξ3 + 2b123 (k1 + k2 + k3)2 eξ1+ξ2+ξ3 , (44)

N2 =a12 (k1 + k2) eξ1+ξ2 + a13 (k1 + k3) eξ1+ξ3 + a23 (k2 + k3) eξ2+ξ3 .+

b123 (k1 + k2 + k3) eξ1+ξ2+ξ3 + k1e
ξ1 + k2e

ξ2 + k3e
ξ3
)
, (45)

D1 =1 + eξ1 + eξ2 + eξ3 + a12e
ξ1+ξ2 + a13e

ξ1+ξ3 + a23e
ξ2+ξ3 + b123e

ξ1+ξ2+ξ3 , (46)

D2 =D1. (47)

Again, without loss of generality, we have set ε = 1 in eqn. (8).

Similarly to the two-soliton solution, we can obtain the dispersion relations by substitut-
ing, in turn, the 1st, 2nd and 3rd terms of f1 into eqn. (10), to obtain

c1 = k2
1, c2 = k2

2, c3 = k2
3. (48)

We now substitute pairs of f1 terms along with the coresponding term of f2 into eqn.
(11), i.e. terms 1 and 2 of f1 plus term 1 of f2 inserted into eqn. (11) yields an expression

that can be solved for a12. Coupling constants a13 and a23 are found in the same way to
yield,

a12 =
(k1 − k2)2

(k1 + k2)2
, a13 =

(k1 − k3)2

(k1 + k3)2
, a23 =

(k2 − k3)2

(k2 + k3)2
. (49)

Similarly, on substituting f1, f2 and f3 into eqn. (12) we obtain an expression which can
be solved for the coupling constant b123 in terms of wave numbers, k1, k2 and k3. This
yields,

b123 =
(k2 − k3)2(k1 − k3)2(k1 − k2)2

(k2 + k3)2(k1 + k3)2(k1 + k2)2
. (50)

We now have a fully defined three-soliton solution for KdV eqn (27). A plot of this
solution is given in Fig (3).

A Maple program that solves this three-soliton example and generates an animation is
given in Listing (3).

Listing 3: Maple program that uses the Hirota method to derive a three-soliton solution to KdV eqn (27)

# Hirota bilinear method

# Three -soliton Solution to KdV equation

restart; with(PDEtools): with(PolynomialTools):

with(plots):

alias(u=u(x,t), v=v(x,t), f=f(x,t)):

# KdV equation

pde1 := diff(u,t)+6*u*diff(u,x)+diff(u,x,x,x);

# Define Hirota Derivatives

Hirota_Dxm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,x$(m-r))

*diff(g,x$r),r=0..m);

Hirota_Dtm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,t$(m-r))

*diff(g,t$r),r=0..m);

Hirota_Dxt :=(f,g)->f*diff(g,x,t)-diff(f,x)*diff(g,t)

-diff(f,t)*diff(g,x)+diff(f,x,t)*g;

v:=2* diff(ln(f),x,x);

declare(f):

pde2:=eval(subs(u=v,pde1))=0;

# Check bilinear transform correct

pde3:= simplify(pde2 ,size);

pde3_chk := simplify(diff(( Hirota_Dxt(f,f)+Hirota_Dxm(f,f,4))/(f^2),x),size)=0;

testeq(lhs(pde3)=lhs(pde3_chk));

# Ansatz for ’f_1 ’

f11:=exp(k[1]*(x-c[1]*t)); f12:=exp(k[2]*(x-c[2]*t));

f13:=exp(k[3]*(x-c[3]*t)); f1:=f11+f12+f13;

B1:=( ff1)->Hirota_Dxt (1,ff1)+Hirota_Dxt(ff1 ,1)+

Hirota_Dxm(ff1 ,1,4)+Hirota_Dxm (1,ff1 ,4)=0;

eqn1a :=B1(f11); # B(1.f1 + f1.1);

eqn1b :=B1(f12); # B(1.f2 + f2.1);

eqn1c :=B1(f13); # B(1.f3 + f3.1);

sol1:=solve(simplify ({eqn1a ,eqn1b ,eqn1c},symbolic) ,{c[1],c[2],c[3]});

assign(sol1);

# Ansatz for ’f2 ’ suggested by equation: B(1.f2+f1.f1+f2.1)=0

f21:=a[12]* combine(f11*f12); f22:=a[13]* combine(f11*f13);

f23:=a[23]* combine(f12*f13); f2:=f21+f22+f23;

B2:=(ff1 ,ff2)->Hirota_Dxt (1,ff2)+Hirota_Dxt(ff2 ,1)+

Hirota_Dxm (1,ff2 ,4)+Hirota_Dxm(ff2 ,1,4)+

Hirota_Dxt(ff1 ,ff1)+Hirota_Dxm(ff1 ,ff1 ,4);

eqn2a := simplify(B2(f11+f12 ,f21),size);

eqn2b := simplify(B2(f11+f13 ,f22),size);

eqn2c := simplify(B2(f12+f13 ,f23),size);

a[12]:= factor(simplify(solve(eqn2a ,a[12]),symbolic));

a[13]:= factor(simplify(solve(eqn2b ,a[13]),symbolic));

a[23]:= factor(simplify(solve(eqn2c ,a[23]),symbolic));

# Ansatz for ’f3 ’ suggested by equation: B(1.f3+f1.f2+f2.f1+f3.1)=0

f3:=b[123]* combine(f11*f12*f13);

# Three -soliton ancilliary function

F:=1+f1+f2+f3;

B3:=(ff1 ,ff2 ,ff3)->Hirota_Dxt (1,ff3)+Hirota_Dxt(ff3 ,1)+

Hirota_Dxm (1,ff3 ,4)+Hirota_Dxm(ff3 ,1,4)+

Hirota_Dxt(ff1 ,ff2)+Hirota_Dxm(ff1 ,ff2 ,4)+

Hirota_Dxt(ff2 ,ff1)+Hirota_Dxm(ff2 ,ff1 ,4);

F2:=B3(f1 ,f2 ,f3):

F3:= solve(F2 ,b[123]):

b[123]:= factor(simplify(F3 ,symbolic)); # b[123]:=a[12]*a[13]*a[23];

# Three -soliton solution

soln:=u=simplify (2* diff(ln(F),x,x),size);

# Check solution - Can take a long time!!

pdetest(soln ,pde1);

# Animate result

zz1:= simplify(subs({k[1]=1,k[2]= sqrt (2),k[3]=2} , rhs(soln)),size);

animate(zz1 ,x=-50..50,t=-10..10,

numpoints =300, frames =100, axes=framed ,

labels =["x","u(x,t)"],thickness =3,

title="KdV Equation: Three -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16]);

Figure 3: Initial animation profile for KdV equation three-soliton solution, eqn. (43), at t = −10, for
k = 1, k2 =

√
2 and k3 = 2. Solitons move left to right with the taller faster solitons overtaking the shorter

slower solitons.

2.4 Four-soliton solution to the KdV equation

This problem is similar to the three-soliton problem except that the auxilliary function
f has an additional term, f4. So now we need to find f1, f2, f3 and f4. Because we are
seeking a four-soliton solution, we use four-term forms of f1, f2 and f3 that were success-
ful for the three-soliton case, i.e. f1 = eξ1 + eξ2 + eξ3 + eξ4 , f2 = a12e

(ξ1+ξ2) + a13e
(ξ1+ξ3) +

a14e
(ξ1+ξ4) + a23e

(ξ2+ξ3) + a24e
(ξ2+ξ4) + a34e

(ξ3+ξ4), f3 = b123e
ξ1+ξ2+ξ3 + b124e

ξ1+ξ2+ξ4 +
b134e

ξ1+ξ3+ξ4 + b234e
ξ2+ξ3+ξ4 along with the ansatz f4 = c123e

ξ1+ξ2+ξ3+ξ4 , where coupling
constants a12, a13, a14, a23, a24, b123, b124, b134, b234 and c1234 are yet to be determined. This
enables us to construct the auxilliary function, f = 1 + εf1 + ε2f2 + ε3f3 + ε4f4.

The ai,j auxillary coupling constants and the bijk auxillary coupling constants are deter-
mined as for the three-soliton case. On substituting in turn all combinations of three
terms from f3, into eqn. (12), i.e. 1st, 2nd and 3rd; 1st, 2nd and 4th; 1st, 3rd and 4th; 2nd,
3rd and 4th; we obtain expressions which can be solved for these coupling constants in

terms of wave numbers, k1, k2, k3 and k4.

Similar to f3 in the three-soliton case, the form for f4 is not an arbitrary choice but,
rather, is suggested by eqn (13) which evaluates to

B(1 · f4 + f4 · 1) +B(f1 · f3 + f2 · f2 + f3 · f1) = B(1 · f4 + f4 · 1)−
6Qeξ1+ξ2+ξ3+ξ4

(k1 + k2)2 (k1 + k3)2 (k1 + k4)2 (k2 + k3)2 (k2 + k4)2 (k3 + k4)2 = 0, (51)

where Q is a complicated expression involving wave numbers, ki, i = 1, · · · , 4.

For eqn (51) to be equal to zero, clearly, the first term must be of the form,

B(1 · f4 + f4 · 1) = c1234e
ξ1+ξ2+ξ3+ξ4 . (52)

We now substitute the auxilliary function f into eqn. (7) to obtain the solution

u =
N1

D1

+ 2

(
N2

D2

)2

, (53)

where N1, N2, D1 and D2 are defined below. A plot of ths solution is given in Fig (4).

N1 =2k1
2eξ1 + 2k2

2eξ2 + 2k3
2eξ3 + 2k4

2eξ4 + 2 (k1 − k2)2 eξ1+ξ2 + 2 (k1 − k3)2 eξ1+ξ3+

2 (k1 − k4)2 eξ1+ξ4 + 2 (k2 − k3)2 eξ2+ξ3 + 2 (k2 − k4)2 eξ2+ξ4 + 2 (k3 − k4)2 eξ3+ξ4+

2b123 (k1 + k2 + k3)2 eξ1+ξ2+ξ3 + 2b124 (k1 + k2 + k4)2 eξ1+ξ2+ξ4+

2b134 (k1 + k3 + k4)2 eξ1+ξ3+ξ4 + 2b234 (k2 + k3 + k4)2 eξ2+ξ3+ξ4+

2c1234 (k1 + k2 + k3 + k4)2 eξ1+ξ2+ξ3+ξ4 . (54)

D1 =1 + eξ1 + eξ2 + eξ3 + eξ4 + a12e
ξ1+ξ2 + a13e

ξ1+ξ3 + a14e
ξ1+ξ4+

a23e
ξ2+ξ3 + a24e

ξ2+ξ4 + a34e
ξ3+ξ4 + b123e

ξ1+ξ2+ξ3 + b124e
ξ1+ξ2+ξ4+

b134e
ξ1+ξ3+ξ4 + b234e

ξ2+ξ3+ξ4 + c1234e
ξ1+ξ2+ξ3+ξ4 . (55)

N2 =k1e
ξ1 + k2e

ξ2 + k3e
ξ3 + k4e

ξ4 +
(k1 − k2)2 eξ1+ξ2

k1 + k2

+
(k1 − k3)2 eξ1+ξ3

k1 + k3

+

(k1 − k4)2 eξ1+ξ4

k1 + k4

+
(k2 − k3)2 eξ2+ξ3

k2 + k3

+
(k2 − k4)2 eξ2+ξ4

k2 + k4

+
(k3 − k4)2 eξ3+ξ4

k3 + k4

+

b123 (k1 + k2 + k3) eξ1+ξ2+ξ3 + b124 (k1 + k2 + k4) eξ1+ξ2+ξ4+

b134 (k1 + k3 + k4) eξ1+ξ3+ξ4 + b234 (k2 + k3 + k4) eξ2+ξ3+ξ4+

c1234 (k1 + k2 + k3 + k4) eξ1+ξ2+ξ3+ξ4 . (56)

D2 =1 + eξ1 + eξ2 + eξ3 + eξ4 + a12e
ξ1+ξ2 + a13e

ξ1+ξ3 + a14e
ξ1+ξ4 + a23e

ξ2+ξ3+

a24e
ξ2+ξ4 + a34e

ξ3+ξ4 + b123e
ξ1+ξ2+ξ3 + b124e

ξ1+ξ2+ξ4 + b134e
ξ1+ξ3+ξ4+

b234e
ξ2+ξ3+ξ4 + c1234e

ξ1+ξ2+ξ3+ξ4 . (57)

Where, again without loss of generality, we have set ε = 1 in eqn. (8).

As for the single-soliton solution, we can obtain the dispersion relations by substituting,
in turn, the four terms of f1 into eqn. (10), to obtain

c1 = k2
1, c2 = k2

2, c3 = k2
3, c4 = k2

4. (58)

The coupling constants a, b and c are determined as described above and yield,

a12 =
(k1 − k2)2

(k1 + k2)2 , a13 =
(k1 − k3)2

(k1 + k3)2 , a14 =
(k1 − k4)2

(k1 + k4)2 . (59)

a23 =
(k2 − k3)2

(k2 + k3)2 , a24 =
(k2 − k4)2

(k2 + k4)2 , a34 =
(k3 − k4)2

(k3 + k4)2 . (60)

b123 =
(k2 − k3)2 (k1 − k3)2 (k1 − k2)2

(k2 + k3)2 (k1 + k3)2 (k1 + k2)2 , b124 =
(k2 − k4)2 (k1 − k4)2 (k1 − k2)2

(k2 + k4)2 (k1 + k4)2 (k1 + k2)2 . (61)

b134 =
(k3 − k4)2 (k1 − k4)2 (k1 − k3)2

(k3 + k4)2 (k1 + k4)2 (k1 + k3)2 , b234 =
(k3 − k4)2 (k2 − k4)2 (k2 − k3)2

(k3 + k4)2 (k2 + k4)2 (k2 + k3)2 . (62)

c1234 =
(k3 − k4)2 (k2 − k4)2 (k2 − k3)2 (k1 − k4)2 (k1 − k3)2 (k1 − k2)2

(k3 + k4)2 (k2 + k4)2 (k2 + k3)2 (k1 + k4)2 (k1 + k3)2 (k1 + k2)2 . (63)

We now have a fully defined four-soliton solution for KdV eqn (27). A plot of this solution
is given in Fig (4).

A Maple program that solves this four-soliton example and generates 2D and 3D anima-
tions, is given in Listing (4).

Listing 4: Maple program that uses the Hirota method to derive a four-soliton solution to KdV eqn (27)

# Hirota bilinear method

# Four -soliton Solution to KdV equation

restart; with(PDEtools): with(PolynomialTools):

with(plots):

alias(u=u(x,t), v=v(x,t), f=f(x,t)):

# KdV equation

pde1 := diff(u,t)+6*u*diff(u,x)+diff(u,x,x,x);

# Define Hirota Derivatives

Hirota_Dxm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,x$(m-r))

*diff(g,x$r),r=0..m);

Hirota_Dtm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,t$(m-r))

*diff(g,t$r),r=0..m);

Hirota_Dxt :=(f,g)->f*diff(g,x,t)-diff(f,x)*diff(g,t)

-diff(f,t)*diff(g,x)+diff(f,x,t)*g;

v:=2* diff(ln(f),x,x);

declare(f):

pde2:=eval(subs(u=v,pde1))=0;

# Check bilinear transform correct

pde3:= simplify(pde2 ,size);

pde3_chk := simplify(diff(( Hirota_Dxt(f,f)+

Hirota_Dxm(f,f,4))/(f^2),x),size)=0;

testeq(lhs(pde3)=lhs(pde3_chk));

# Ansatz for ’f_1 ’

f11:=exp(k[1]*(x-c[1]*t)); f12:=exp(k[2]*(x-c[2]*t));

f13:=exp(k[3]*(x-c[3]*t)); f14:=exp(k[4]*(x-c[4]*t));

f1:=f11+f12+f13+f14;

B1:=( ff1)->Hirota_Dxt (1,ff1)+Hirota_Dxt(ff1 ,1)+

Hirota_Dxm(ff1 ,1,4)+Hirota_Dxm (1,ff1 ,4)=0;

eqn1a :=B1(f11); # B(1.f1 + f1.1);

eqn1b :=B1(f12); # B(1.f2 + f2.1);

eqn1c :=B1(f13); # B(1.f3 + f3.1);

eqn1d :=B1(f14); # B(1.f3 + f4.1);

sol1:=solve(simplify ({eqn1a ,eqn1b ,eqn1c ,eqn1d},symbolic),

{c[1],c[2],c[3],c[4]});

assign(sol1);

# Ansatz for ’f2 ’ suggested by equation: B(1.f2+f1.f1+f2.1)=0

f21:=a[12]* combine(f11*f12); f22:=a[13]* combine(f11*f13);

f23:=a[14]* combine(f11*f14); f24:=a[23]* combine(f12*f13);

f25:=a[24]* combine(f12*f14); f26:=a[34]* combine(f13*f14);

f2:=f21+f22+f23+f24+f25+f26;

B2:=(ff1 ,ff2)->Hirota_Dxt (1,ff2)+Hirota_Dxt(ff2 ,1)+

Hirota_Dxm (1,ff2 ,4)+Hirota_Dxm(ff2 ,1,4)+

Hirota_Dxt(ff1 ,ff1)+Hirota_Dxm(ff1 ,ff1 ,4);

eqn2a := simplify(B2(f11+f12 ,f21),size);

eqn2b := simplify(B2(f11+f13 ,f22),size);

eqn2c := simplify(B2(f11+f14 ,f23),size);

eqn2d := simplify(B2(f12+f13 ,f24),size);

eqn2e := simplify(B2(f12+f14 ,f25),size);

eqn2f := simplify(B2(f13+f14 ,f26),size);

a[12]:= factor(simplify(solve(eqn2a ,a[12]),symbolic));

a[13]:= factor(simplify(solve(eqn2b ,a[13]),symbolic));

a[14]:= factor(simplify(solve(eqn2c ,a[14]),symbolic));

a[23]:= factor(simplify(solve(eqn2d ,a[23]),symbolic));

a[24]:= factor(simplify(solve(eqn2e ,a[24]),symbolic));

a[34]:= factor(simplify(solve(eqn2f ,a[34]),symbolic));

# Ansatz for ’f3 ’ suggested by equation: B(1.f3+f1.f2+f2.f1+f3.1)=0

f31:=b[123]* combine(f11*f12*f13);

f32:=b[124]* combine(f11*f12*f14);

f33:=b[134]* combine(f11*f13*f14);

f34:=b[234]* combine(f12*f13*f14); f3:=f31+f32+f33+f34;

B3:=(ff1 ,ff2 ,ff3)->Hirota_Dxt (1,ff3)+Hirota_Dxt(ff3 ,1)+

Hirota_Dxm (1,ff3 ,4)+Hirota_Dxm(ff3 ,1,4)+

Hirota_Dxt(ff1 ,ff2)+Hirota_Dxm(ff1 ,ff2 ,4)+

Hirota_Dxt(ff2 ,ff1)+Hirota_Dxm(ff2 ,ff1 ,4);

eqn3a := simplify(B3(f11+f12+f13 ,f21+f22+f24 ,f31),symbolic):

eqn3b := simplify(B3(f11+f12+f14 ,f21+f23+f25 ,f32),symbolic):

eqn3c := simplify(B3(f11+f13+f14 ,f22+f23+f26 ,f33),symbolic):

eqn3d := simplify(B3(f12+f13+f14 ,f24+f25+f26 ,f34),symbolic):

#t:=0;x:=0;

sol_123 := factor(simplify(solve ({eqn3a},{b[123]}) ,size));

sol_124 := factor(simplify(solve ({eqn3b},{b[124]}) ,size));

sol_134 := factor(simplify(solve ({eqn3c},{b[134]}) ,size));

sol_234 := factor(simplify(solve ({eqn3d},{b[234]}) ,size));

assign(sol_123 ,sol_124 ,sol_134 ,sol_234);

# Ansatz for ’f4 ’ suggested by equation: B(1.f4+f1.f3+f2.f2+f3.f1+f4.1)=0

f4:=c[1234]* combine(f11*f12*f13*f14);

F:=1+f1+f2+f3+f4;

B4:=(ff1 ,ff2 ,ff3 ,ff4)->Hirota_Dxt (1,ff4)+Hirota_Dxm (1,ff4 ,4)+

Hirota_Dxt(ff4 ,1)+Hirota_Dxm(ff4 ,1,4)+

Hirota_Dxt(ff1 ,ff3)+Hirota_Dxm(ff1 ,ff3 ,4)+

Hirota_Dxt(ff3 ,ff1)+Hirota_Dxm(ff3 ,ff1 ,4)+

Hirota_Dxt(ff2 ,ff2)+Hirota_Dxm(ff2 ,ff2 ,4);

F2:=B4(f1 ,f2 ,f3 ,f4):

F3:= solve(F2 ,c[1234]):

# c[1234]:=a[12]*a[13]*a[14]*a[23]*a[24]*a[34];

c[1234]:= factor(simplify(F3 ,symbolic));

# Four -soliton solution

soln:=u=2* diff(ln(F),x,x);

# Animate result

zz1:= simplify(subs({k[1]=1,k[2]= sqrt (2),k[3]=2,k[4]=3} , rhs(soln)),size):

animate(zz1 ,x=-50..50,t=-5..5,

numpoints =300, frames =30, axes=framed ,

labels =["x","u(x,t)"],thickness =3,

title="KdV Equation: Four -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16]);

animate3d(zz1 ,x= -50..50 ,y= -50..50 ,t=-5..5,

frames =10, axes=framed , grid =[100 ,100] ,

gridstyle=triangular ,style=patchnogrid , shading=ZHUE ,

title ="1D-KdV Equation: Four -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16], labels =["x","y","u(x,t)"],

orientation =[55 ,55]);

#Check Solution - takes a long time on Maple 16 !!

pdetest(soln ,pde1);

Figure 4: Initial animation profile for KdV equation four-soliton solution, eqn. (53), at t = −5, for k = 1,
k2 =

√
2, k3 = 2 and k4 = 3. Solitons move left to right with the taller faster solitons overtaking the shorter

slower solitons.

2.5 Two-Soliton solution to the Boussinesq equation

utt − uxx − 3(u2)xx − u4x = 0. (64)

If we substitute eqn. (7) into eqn. (64) we obtain,

1

f 4

[
−2f6xf

3 + 12f5xfxf
2 +

(
−36fx

2f − 2f 3 + 6fxxf
2
)
f4x (65)

+ 2fttxxf
3 − 4f3x

2f 2 +
(
48fx

3 + 8fxf
2
)
f3x− 4fxfttxf

2 − 4ftxxftf
2

+ 12fxx
3f +

(
6f 2 − 36fx

2
)
fxx

2 − 2f
(
12fx

2 − 2ft
2 + fttf

)
fxx

+ 16fxftxftf − 12fx
2ft

2 + 12fx
4 − 4ftx

2f 2 + 4fx
2fttf

]
= 0.

Then, using Hirota’s bilinear operator, the result can be expressed in the following suc-
cinct bilinear form

∂2

∂x2

[
1

f 2

(
D2
t −D2

x −D4
x

)
(f · f)

]
= 0. (66)

Hence, integrating twice with respect to x and taking the integration functions to be
zero, the bilinear equation equivalent to eqn (64) is

B(f · f) =
(
D2
t −D2

x −D4
x

)
(f · f) = 0. (67)

Thus, if f is a solution of eqn. (67), then u from eqn. (7) is a solution to eqn. (64). See
[Rog-82, p69-71] for a detailed discussion and derivation.

If we now follow the same procedure as for the two-soliton solution to the KdV equation
we obtain the following solution,

u =
2
[(
k2

2eξ1 + k1
2eξ2 + (k1 + k2)2) a12e

ξ1+ξ2 +
(
(k1 − k2)2 eξ2 + k1

2
)
eξ1 + k2

2eξ2
]

(1 + eξ1 + eξ2 + a12eξ1+ξ2)2 .

(68)

As for the KdV two-soliton solution, we obtain the dispersion relations by substituting,
in turn, the two terms of f1 into eqn. (10), to obtain

c1 = ±
√

1 + k1
2, c2 = ±

√
1 + k2

2 (69)

We choose the positive solutions.

Then on substituting f1 and f2 into eqn. (11) we obtain the following expression for the
coupling constant,

a12 =

√
1 + k1

2
√

1 + k2
2 − 1− 2k1

2 − 2k2
2 + 3k1k2√

1 + k1
2
√

1 + k2
2 − 1− 2k1

2 − 3k1k2 − 2k2
2
. (70)

We now have a fully defined two-soliton solution to Boussinesq eqn (64). A plot of this
solution is given in Fig (5).

A Maple program that solves this two-soliton example and generates an animation is
given in Listing (5).

Listing 5: Maple program that uses the Hirota method to derive a two-soliton solution to Boussinesq eqn
(64)

# Hirota bilinear method

# Two -soliton Solution to Boussinesq equation

restart; with(PDEtools): with(PolynomialTools):

with(plots):

alias(u=u(x,t), v=v(x,t), f=f(x,t)):

# Bousinesq equation

pde1 := diff(u,t,t)-diff(u,x,x) -3*diff(u^2,x,x)-diff(u,x,x,x,x);

# Define Hirota Derivatives

Hirota_Dxm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,x$(m-r))

*diff(g,x$r),r=0..m);

Hirota_Dtm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,t$(m-r))

*diff(g,t$r),r=0..m);

Hirota_Dxt :=(f,g)->f*diff(g,x,t)-diff(f,x)*diff(g,t)

-diff(f,t)*diff(g,x)+diff(f,x,t)*g;

v:=2* diff(ln(f),x,x);

declare(f):

pde2:=eval(subs(u=v,pde1))=0;

# Check bilinear transform correct

pde3:= simplify(pde2 ,size);

pde3_chk := simplify(diff(( Hirota_Dtm(f,f,2)-Hirota_Dxm(f,f,2)-

Hirota_Dxm(f,f,4))/f^2,x,x),size)=0;

testeq(lhs(pde3)=lhs(pde3_chk));

# Ansatz for ’f_1 ’

f11:=exp(k[1]*(x-c[1]*t)); f12:=exp(k[2]*(x-c[2]*t));

f1:=f11+f12;

B1:=( ff1)->Hirota_Dtm (1,ff1 ,2)+Hirota_Dtm(ff1 ,1,2)-

(Hirota_Dxm (1,ff1 ,2)+Hirota_Dxm(ff1 ,1,2))-

(Hirota_Dxm (1,ff1 ,4)+Hirota_Dxm(ff1 ,1,4))=0;

eqn1a :=B1(f11); # B(1.f1 + f1.1);

eqn1b :=B1(f12); # B(1.f2 + f2.1);

sol1:=solve(simplify ({eqn1a ,eqn1b},symbolic) ,{c[1],c[2]});

# Note: Boussinesq eqn has +/- velocity values.

# Therefore waves can travel in opposite directions!

sol11 := allvalues(sol1 [1]);sol12:= allvalues(sol1 [2]);

assign(sol11 [1],sol12 [1]); c[1];c[2];

# Ansatz for ’f2 ’ suggested by equation: B(1.f2+f1.f1+f2.1)=0

f2:=a[12]* combine(f11*f12);

# Two -soliton ancilliary function

F:=1+f1+f2;

B2:=(ff1 ,ff2) ->(Hirota_Dtm (1,ff2 ,2)+Hirota_Dtm(ff2 ,1,2))-

(Hirota_Dxm (1,ff2 ,2)+Hirota_Dxm(ff2 ,1,2))-

(Hirota_Dxm (1,ff2 ,4)+Hirota_Dxm(ff2 ,1,4))+

(Hirota_Dtm(ff1 ,ff1 ,2)-Hirota_Dxm(ff1 ,ff1 ,2) -

Hirota_Dxm(ff1 ,ff1 ,4));

f1;f2;

eqn2:= simplify(B2(f1,f2),size)=0;

a[12]:= factor(simplify(solve(eqn2 ,a[12]),symbolic));

# Two -soliton solution

soln:=u=simplify (2* diff(ln(F),x,x),size);;

# Animate result

zz1:= simplify(subs({k[1]=1,k[2]=1/2} , rhs(soln)),symbolic);

animate(zz1 ,x= -200..150 ,t= -130..95 ,

numpoints =1000 , frames =50, axes=framed ,

labels =["x","u(x,t)"],thickness =3,

title=" Boussinesq Equation: Two -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16]);

# Check Solution - takes a long time with Maple 16

simplify(subs(soln ,pde1),symbolic);

Figure 5: Initial animation profile for Boussinesq equation two-soliton solution, eqn (68), at t = −130, for
k = 1 and k2 = 1/2. Solitons move left to right with the taller faster soliton overtaking the shorter slower
soliton.

2.6 Two-Soliton solution to the Sawada-Kotera equation

The Sawada-Kotera (SK) equation is given by,

ut + 45u2ux + 15uxuxx + 15uu3x + u5x = 0. (71)

This equation is also known as the Caudrey-Dodd-Gibbon equation (CDGE).

To demonstrate that the bilinear form is correct, we derive the potential form of the
Sawada-Kotera equation by substituting u = ∂w/∂x into eqn (71) to obtain

wx,t + 45wx
2wx,x + 15wx,xw3x + 15wxw4x + w6x = 0. (72)

Also, we note that eqn (7) becomes

w(x, t) = 2
∂ ln f(x, t)

∂x
= 2

fx
f

(73)

which, on substituting into eqn (72), we obtain

−2fxft − 12f5xfx + 30f4xfx,x + 2ft,xf + 2f6xf − 20f3x
2

f 2
= 0. (74)

This equation can be written more succinctly in the following form using Hirota’s bilinear
operator

1

f 2

(
DxDt +D6

x

)
(f · f) = 0. (75)

Hence the bilinear equation equivalent to eqns (71) and (72) is [Dra-92]

B(f · f) =
(
DxDt +D6

x

)
(f · f) = 0. (76)

Thus, if f is a solution of eqn. (76), then w from eqn. (73) is a solution to eqn. (72) and
u from eqn. (7) is a solution to eqn. (71).

If we now follow the same procedure as for the two-soliton solution to the KdV equation
we obtain the following solution,

u =
2
((
k2

2eξ1 + k1
2eξ2 + (k1 + k2)2) a12e

ξ1+ξ2 +
(
(k1 − k2)2 eξ2 + k1

2
)
eξ1 + k2

2eξ2
)

(1 + eξ1 + eξ2 + a12eξ1+ξ2)2 .

(77)

which has the same form as the two-soliton solution to the Boussinesq equation. How-
ever, the Sawada-Kotera equation has different dispersion relations which we obtain by
substituting, in turn, the two terms of f1 into eqn. (10), to obtain

c1 = k4
1, c2 = k4

2. (78)

Also, on substituting f1 and f2 into eqn. (11) we obtain the following expression for the
coupling constant,

a12 =

(
k1

2 − k1k2 + k2
2
)

(k1 − k2)2(
k1

2 + k1k2 + k2
2
)

(k1 + k2)2 . (79)

Again, this coupling constant is different to that of the Boussinesq eqn. We now have a
fully defined two-soliton solution to Sawada-Kotera eqn (71). A plot of this solution is
given in Fig (6).

A Maple program that solves this two-soliton example and generates an animation is
given in Listing (6).

Listing 6: Maple program that uses the Hirota method to derive a two-soliton solution to Sawada-Kotera
eqn (71)

# Hirota bilinear method

# Two -soliton Solution to Sawada -Kotera equation

restart; with(PDEtools): with(PolynomialTools):

with(plots):

alias(u=u(x,t), v=v(x,t), w=w(x,t), f=f(x,t)):

# SW equation

pde1 := diff(u,t)+45*u^2* diff(u,x)+15* diff(u,x)*diff(u,x,x) +

15*u*diff(u,x,x,x)+diff(u,x,x,x,x,x)=0;

eqn1:=u=diff(w,x);

declare(w);

# Derive potential version of pde1

pde1a :=subs(eqn1 ,pde1);

# manually integrate pde1a

pde1b :=diff(w, t)+15* diff(w, x)^3+15* diff(w, x)*diff(w, x, x, x)+diff(w, x, x,

x, x, x) = 0;

# Differentiate pde1b

pde1c :=diff(pde1b ,x);

# Check that integration is correct

testeq(lhs(pde1a)=lhs(pde1c));

# Define Hirota Derivatives

Hirota_Dxm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,x$(m-r))

*diff(g,x$r),r=0..m);

Hirota_Dtm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,t$(m-r))

*diff(g,t$r),r=0..m);

Hirota_Dxt :=(f,g)->f*diff(g,x,t)-diff(f,x)*diff(g,t)

-diff(f,t)*diff(g,x)+diff(f,x,t)*g;

# Hirota logrithmic transformation (Cole -Hopf)

lnFunc :=w=2* diff(ln(f),x); # equivalent to: u:=2* diff(ln(f),x,x)

declare(f):

pde2:=eval(subs(lnFunc ,pde1b));

# Check bilinear transformation is correct

pde3:= simplify(pde2 ,size);

pde3_chk := normal(Hirota_Dxt(f,f)+Hirota_Dxm(f,f,6))/f^2=0;

testeq(lhs(pde3)-lhs(pde3_chk));

# Ansatz for ’f_1 ’

f11:=exp(k[1]*(x-c[1]*t)); f12:=exp(k[2]*(x-c[2]*t));

f1:=f11+f12;

# B1=B((1.f + f.1))

B1:=( ff1)->Hirota_Dxt (1,ff1)+Hirota_Dxt(ff1 ,1)+

Hirota_Dxm(ff1 ,1,6)+Hirota_Dxm (1,ff1 ,6)=0;

eqn1a :=B1(f11); # B(1.f1 + f1.1);

eqn1b :=B1(f12); # B(1.f2 + f2.1);

sol1:=solve(simplify ({eqn1a ,eqn1b},symbolic) ,{c[1],c[2]});

assign(sol1); # ==>> Dispersion relations

# Ansatz for ’f2 ’ suggested by equation: B(1.f2+f1.f1+f2.1)=0

f2:=a[12]* combine(f11*f12);

# Two -soliton ancilliary function

F:=1+f1+f2;

B2:=(ff1 ,ff2)->Hirota_Dxt (1,ff2)+Hirota_Dxt(ff2 ,1)+

Hirota_Dxm (1,ff2 ,6)+Hirota_Dxm(ff2 ,1,6)+

Hirota_Dxt(ff1 ,ff1)+Hirota_Dxm(ff1 ,ff1 ,6);

eqn2:= simplify(B2(f1,f2),size);

a[12]:= factor(simplify(solve(eqn2 ,a[12]),symbolic));

# Two -soliton solution

soln:=u=simplify (2* diff(ln(F),x,x),size);;

# Animate result

zz1:= simplify(subs({k[1]=1,k[2]=1/2} , rhs(soln)),symbolic);

animate(zz1 ,x=-50..50,t=-30..30,

numpoints =1000 , frames =50, axes=framed ,

labels =["x","u(x,t)"],thickness =3,

title="Sawada -Kotera Equation: Two -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16]);

# Check Solution

pdetest(soln ,pde1);

2.7 Two-Soliton solution to the 2D Kadomtsev-Petviashvili equation

The 2D Kadomtsev-Petviashvili (KP) equation is given by,

(ut + 6uux + u3x)x ± 3uyy = 0. (80)

If we substitute eqn. (7) into eqn. (80) we obtain

−2

f 4

[
−6fx

2ft,xf + 6fx
3ft − 18f4xfx

2f + 24f3xfx
3 − 18fxx

2fx
2

− 2fxxfy
2f + 6fx

2fy
2 + 6f5xfxf

2 + 3f4xfxxf
2 + 2fxxyfyf

2

+ fxxfy,yf
2 + 2fxfxyyf

2 − 2fx
2fyyf + 3ftxxfxf

2 + 3fxxftxf
2

+ f3xftf
2 − 6fxxftfxf − 8fxfxyfyf − 2f3x

2f 2 + 6fxx
3f

+2fxy
2f 2 − ft3xf 3 − f6xf

3 − fxxyyf 3
]

= 0 (81)

Figure 6: Initial animation profile for Sawada-Kotera equation two-soliton solution, eqn. (77), at t = −30,
for k = 1 and k2 = 1/2. Solitons move left to right with the taller faster soliton overtaking the shorter
slower soliton.

Then, using Hirota’s bilinear operator, the result can be expressed in the following suc-
cinct bilinear form

∂2

∂x2

[
1

f 2

(
DxDt +D4

x ± 3D2
y

)
(f · f)

]
= 0. (82)

Hence, integrating twice with respect to x and taking the integration functions to be
zero, the bilinear equation equivalent to eqn. (80) is [Her-91]

B(f · f) =
(
DxDt +D4

x ± 3D2
y

)
(f · f) = 0. (83)

Thus, if f is a solution of eqn. (83), then u from eqn. (7) is a solution to eqn. (80).

In our analysis we elect to use +3uyy in eqn. (80) and +3D2
y in eqn. (83).

This problem is the same as the KdV two-soliton problem except that it is two-dimensional
and has a different bilinear equation. Because we are seeking a two-soliton solution,
we use the two-term form of f1 that was successful for the KdV two-soliton example,
i.e. f1 = eξ1 + eξ2 , along with the ansatz f2 = a12e

(ξ1+ξ2), where a12 is a coupling
constant yet to be determined. This enables us to construct the auxilliary function,
f = 1 + εf1 + ε2f2. However, because we are dealing with a 2D system, we use the
transformations ξ1 = k1x+ l1x− ω1t and ξ2 = k2x+ l2x− ω2t.

Again, the form for f2 is not an arbitrary choice but, rather, is suggested by eqn (11)
which evaluates to

B(1 · f2 + f2.1) +B(f1 · f1) = B(1 · f2 + f2 · 1)−
6 (k2l1 + k2k

2
1 − k1k

2
2 − k1l2) (−k2l1 + k2k

2
1 − k1k

2
2 + k1l2)

k1k2

eξ1+ξ2 = 0. (84)

For eqn (84) to be equal to zero, clearly, the first term must be of the form,

B(1 · f2 + f2 · 1) = a12e
ξ1+ξ2 . (85)

We now substitute the auxilliary function f into eqn. (7) to obtain the solution

u =
2
[
a12

(
k2

1e
ξ2 + k2

2e
ξ1 + (k1 + k2)2

)
eξ1+ξ2 +

(
(k1 − k2)2eξ2 + k2

1)eξ1 + k2
2e
ξ2
)]

(a12eξ1+ξ2 + eξ1 + eξ2 + 1)2 . (86)

Where, again without loss of generality, we have set ε = 1.

As for the KdV two-soliton solution, we can obtain the dispersion relations by substitut-
ing, in turn, terms 1 and 2 of f1 into eqn. (10), to obtain

ω1 =
k4

1 + 3l21
k1

, ω2 =
k4

2 + 3l22
k2

. (87)

Also, on substituting f1 and f2 into eqn. (11) we obtain an expression which can be
solved for the coupling constant in terms of wave numbers, k1 and k2. This yields,

a12 =
(k2l1 + k2k

2
1 − k1k

2
2 − k1l2) (−k2l1 + k2k

2
1 − k1k

2
2 + k1l2)

(−k2l1 + k2k2
1 + k1k2

2 + k1l2) (k2l1 + k2k2
1 + k1k2

2 − k1l2)
. (88)

We now have a fully defined two-soliton solution to the 2D KP eqn (80). A plot of this
solution is given in Fig (7).

A Maple program that solves this two-soliton example and generates an animation is
given in Listing (7).

Listing 7: Maple program that uses the Hirota method to derive a two-soliton solution to 2D Kadomtsev-
Petviashvili eqn (80)

# Hirota bilinear method

# Two -soliton Solution to 2D Kadomtsev -Petviashvili (KP) equation

restart; with(PDEtools): with(PolynomialTools):

with(plots):

alias(u=u(x,y,t), f=f(x,y,t)):

# KP equation

pde1 := diff(diff(u,t)+6*u*diff(u,x)+diff(u,x,x,x),x)+diff(u,y,y);

# Define Hirota Derivatives

Hirota_Dxm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,x$(m-r))

*diff(g,x$r),r=0..m);

Hirota_Dym :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,y$(m-r))

*diff(g,y$r),r=0..m);

Hirota_Dtm :=(f,g,m)->sum((-1)^r*binomial(m,r)*diff(f,t$(m-r))

*diff(g,t$r),r=0..m);

Hirota_Dxt :=(f,g)->f*diff(g,x,t)-diff(f,x)*diff(g,t)

-diff(f,t)*diff(g,x)+diff(f,x,t)*g;

u1:=2* diff(ln(f),x,x); # logrithmic transformation

declare(f):

pde2:=eval(subs(u=u1,pde1))=0:

# Check bilinear transform correct

pde3:= simplify(pde2 ,symbolic);

pde3_chk := simplify(diff(( Hirota_Dxt(f,f)+Hirota_Dxm(f,f,4) +

Hirota_Dym(f,f,2))/f^2,x,x),symbolic)=0;

testeq(lhs(pde3)=lhs(pde3_chk));

# Ansatz for ’f_1 ’

f11:=exp(k[1]*x+l[1]*y-omega [1]*t); f12:=exp(k[2]*x+l[2]*y-omega [2]*t);

f1:=f11+f12;

B1:=( ff1)->Hirota_Dxt (1,ff1)+Hirota_Dxt(ff1 ,1)+

Hirota_Dxm(ff1 ,1,4)+Hirota_Dxm (1,ff1 ,4)+

3*( Hirota_Dym (1,ff1 ,2)+Hirota_Dym(ff1 ,1,2))=0;

eqn1a :=B1(f11); # B(1.f1 + f1.1);

eqn1b :=B1(f12); # B(1.f2 + f2.1);

sol1:=solve(simplify ({eqn1a ,eqn1b},symbolic) ,{omega [1],omega [2]});

assign(sol1); # ==>> Dispersion relations

# Ansatz for ’f2 ’ suggested by equation: B(1.f2+f1.f1+f2.1)=0

f2:=a[12]* combine(f11*f12);

# Two -soliton ancilliary function

F:=1+f1+f2;

B2:=(ff1 ,ff2)->Hirota_Dxt (1,ff2) +Hirota_Dxt(ff2 ,1)+

Hirota_Dxm (1,ff2 ,4)+Hirota_Dxm(ff2 ,1,4)+

3*( Hirota_Dym (1,ff2 ,2)+Hirota_Dym(ff2 ,1,2))+

(a) Top view, t = −20 (b) Side view, t = 0

Figure 7: Profiles of 2D Kadomtsev-Petviashvili equation two-soliton solution, eqn. (86), for k = 1,
k2 =

√
2, l1 = 1/5 and l2 = 5/6. Solitons move right to left with the taller right hand soliton overtaking

the smaller left hand soliton.

Hirota_Dxt(ff1 ,ff1)+Hirota_Dxm(ff1 ,ff1 ,4)+

3* Hirota_Dym(ff1 ,ff1 ,2);

eqn2:= simplify(B2(f1,f2),size)=0;

a[12]:= factor(simplify(solve(eqn2 ,a[12]),symbolic));

# Two -soliton solution

soln:=u=simplify (2* diff(ln(F),x,x),size);;

# Animate result

zz1:=subs({k[1]=1,k[2]= sqrt (2),l[1]=1/5 ,l[2]=5/6} , rhs(soln));

animate3d(zz1 ,x= -50..50 ,y= -50..50 ,t= -20..20 ,

frames =10, axes=framed , grid =[100 ,100] ,

gridstyle=triangular ,style=patchnogrid , shading=ZHUE ,

title ="2D-KP Equation: Two -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16], labels =["x","y","u(x,t)"],

orientation =[90 ,0]);

animate3d(zz1 ,x= -50..50 ,y= -50..50 ,t= -20..20 ,

frames =10, axes=framed , grid =[100 ,100] ,

gridstyle=triangular ,style=patchnogrid , shading=ZHUE ,

title ="2D-KP Equation: Two -soliton solution",

labelfont =[TIMES , ROMAN , 16], axesfont =[TIMES , ROMAN , 16],

titlefont =[TIMES , ROMAN , 16], labels =["x","y","u(x,t)"],

orientation =[112 ,81]);

# Check Solution - takes a long time with Maple 16 !!

# simplify(subs(soln ,pde1),symbolic);

